

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
				Monarch2090
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	Warteschleife setzt auf 0
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife (vorher 0, nachher 0)
		#define	ZÄHLER2 rC	für Nachstart-Warteschleife
		#define	TASTENFLAG rB	für Nachstart-Warteschleife
		#define	ZAHL09 rD	Zufallszahl über RND
		#define	TASTE r9	Rote Taste zum Nachstart
		#define	STATUS rA	Wenn Sonderspiele, dann 2, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
0x01	F4C		DISP #4,GELD_PF_EINER	
0x02	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x03	F04		DZHX	
0x04	F4C	start	DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0x05	1E5		MOVI #E,DELTA_D	-20 Pf
0x06	BE8		CALL DeltaSub	
0x07	9FF		CMPI #F,GELD_DM_ZEHNER	Münzspeicher unter 0?
0x08	E00		BRZ start0	dann zurück zum Anfang
0x09	BF1		CALL Warteschleife	
0x0A	FEA		DOT STATUS	Keine Sonderspiele?
0x0B	E10		BRZ StartSpiel	dann direkt Spiel starten
0x0C	F0E		EXRM	
0x0D	F2D		DISP #2,S_EINER	Sonderspiele-Zähler anzeigen
0x0E	BF1		CALL Warteschleife	
0x0F	F0E		EXRM	
0x10	0D5	StartSpiel	MOV GELD_PF_ZEHNER,DELTA_D	Münzspeicher sichern
0x11	0E6		MOV GELD_DM_EINER,DELTA_E	
0x12	0F7		MOV GELD_DM_ZEHNER,DELTA_F	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x13	10B		MOVI #0,TASTENFLAG	
0x14	F50		DISP #5,RO	Walzen darstellen
0x15	BF8	stopL	CALL HoleZahl	Linke Walze ermitteln
0x16	1F4		MOVI #F,LO	König
0x17	1D3		MOVI #D,LU	2 DM
0x18	98D		CMPI #8,ZAHL09	
0x19	D2E		BRC L_fertig	
0x1A	1E4		MOVI #E,LO	Krone
0x1B	193		MOVI #9,LU	1 DM
0x1C	97D		CMPI #7,ZAHL09	
0x1D	D2E		BRC L_fertig	
0x1E	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x1F	96D		CMPI #6,ZAHL09	
0x20	D2E		BRC L_fertig	
0x21	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x22	95D		CMPI #5,ZAHL09	
0x23	D2E		BRC L_fertig	
0x24	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x25	94D		CMPI #4,ZAHL09	
0x26	D2E		BRC L_fertig	
0x27	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x28	93D		CMPI #3,ZAHL09	
0x29	D2E		BRC L_fertig	
0x2A	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x2B	91D		CMPI #1,ZAHL09	
0x2C	D2E		BRC L_fertig	
0x2D	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x2E	71C	L_fertig	SUBI #1,ZÄHLER2	ZÄHLER2 enthält zu Beginn 0
0x2F	E35		BRZ stopR	Wenn Wartezeit vorbei, dann rechte Walze
0x30	FD9		DIN TASTE	
0x31	E2E		BRZ L_fertig	Wenn keine Taste, dann Schleifenanfang
0x32	58B		ADDI #8,TASTENFLAG	Wurde Taste zum 2. Mal gedrückt?
0x33	D35		BRC stopR	Dann Taste ignorieren und rechte Walze
0x34	C15		GOTO stopL	Sonst linke Walze nachstarten
0x35	BF8	stopR	CALL HoleZahl	Rechte Walze ermitteln
0x36	1F0		MOVI #F,RO	König
0x37	1D1		MOVI #D,RU	2 DM
0x38	98D		CMPI #8,ZAHL09	
0x39	D4B		BRC R_fertig	
0x3A	1E0		MOVI #E,RO	Krone
0x3B	191		MOVI #9,RU	1 DM
0x3C	97D		CMPI #7,ZAHL09	
0x3D	D4B		BRC R_fertig	
0x3E	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3F	96D		CMPI #6,ZAHL09	
0x40	D4B		BRC R_fertig	
0x41	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x42	95D		CMPI #5,ZAHL09	
0x43	D4B		BRC R_fertig	
0x44	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x45	93D		CMPI #3,ZAHL09	
0x46	D4B		BRC R_fertig	
0x47	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x48	91D		CMPI #1,ZAHL09	
0x49	D4B		BRC R_fertig	
0x4A	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x4B	BF1	R_fertig	CALL Warteschleife	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x4C	BF8	stopM	CALL HoleZahl	Mittlere Walze ermitteln
0x4D	1F2		MOVI #F,MI	König
0x4E	98D		CMPI #8,ZAHL09	
0x4F	D5D		BRC M_fertig	
0x50	1C2		MOVI #C,MI	1,60 DM
0x51	97D		CMPI #7,ZAHL09	
0x52	D5D		BRC M_fertig	
0x53	142		MOVI #4,MI	40 Pf (2x)
0x54	95D		CMPI #5,ZAHL09	
0x55	D5D		BRC M_fertig	
0x56	182		MOVI #8,MI	80 Pf (2x)
0x57	93D		CMPI #3,ZAHL09	
0x58	D5D		BRC M_fertig	
0x59	1A2		MOVI #A,MI	1,20 DM (2x)
0x5A	91D		CMPI #1,ZAHL09	
0x5B	D5D		BRC M_fertig	
0x5C	1D2		MOVI #D,MI	2 DM (2x)
0x5D	BF1	M_fertig	CALL Warteschleife	
0x5E	05D		MOV DELTA_D,GELD_PF_ZEHNER	Münzspeicher wiederherstellen
0x5F	06E		MOV DELTA_E,GELD_DM_EINER	
0x60	07F		MOV DELTA_F,GELD_DM_ZEHNER	
				Gewinnabtastung
0x61	109		MOVI #0,CHANCE2	
0x62	10B		MOVI #0,GEWINNSYMBOL	
0x63	BFC		CALL LöscheDelta	
0x64	840	ChanceA	CMP LO,RO	
0x65	E69		BRZ erste_Chance	
0x66	841		CMP LO,RU	
0x67	E69		BRZ erste_Chance	
0x68	C6A		GOTO ChanceB	
0x69	048	erste_Chance	MOV LO,CHANCE1	
0x6A	830	ChanceB	CMP LU,RO	
0x6B	E6F		BRZ zweite_Chance	
0x6C	831		CMP LU,RU	
0x6D	E6F		BRZ zweite_Chance	
0x6E	C70		GOTO Mitte	
0x6F	039	zweite_Chance	MOV LU,CHANCE2	
0x70	9E2	Mitte	CMPI #E,MI	König (Joker) in der Mitte?
0x71	D7B		BRC Joker_ja	
0x72	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x73	E75		BRZ gewinn1	
0x74	C76		GOTO test2	
0x75	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x76	829	test2	CMP MI,CHANCE2	
0x77	E79		BRZ gewinn2	
0x78	C82		GOTO symbol	
0x79	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x7A	C82		GOTO symbol	
0x7B	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x7C	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x7D	D7F		BRC aaa	
0x7E	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x7F	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x80	D82		BRC symbol	
0x81	09B		MOV CHANCE2,GEWINNSYMBOL	bester Gewinn

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x82	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x83	E86		BRZ G100	
0x84	D87		BRC Gewinn120	
0x85	C96		GOTO GewinnBis80Pf	
0x86	1A5	G100	MOVI #A,DELTA_D	Hex 10
0x87	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x88	E8A		BRZ G120	
0x89	C8B		GOTO Gewinn140	
0x8A	1C5	G120	MOVI #C,DELTA_D	Hex 12
0x8B	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x8C	E8E		BRZ G140	
0x8D	C8F		GOTO Gewinn160	
0x8E	1E5	G140	MOVI #E,DELTA_D	Hex 14
0x8F	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x90	E93		BRZ G160	
0x91	DA0		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x92	C97		GOTO SoSpTest	
0x93	105	G160	MOVI #0,DELTA_D	Hex 16
0x94	116		MOVI #1,DELTA_E	Hex 16
0x95	C97		GOTO SoSpTest	
0x96	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x97	90A	SoSpTest	CMPI #0,STATUS	Laufen gerade keine Sonderspiele?
0x98	EA2		BRZ GongTest	
0x99	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x9A	DA0		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9B	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x9C	EA0		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9D	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x9E	DA0		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9F	CA8		GOTO GewinnAnzeige	
0xA0	145	Erhöhung	MOVI #4,DELTA_D	2 DM (Hex 20)
0xA1	116		MOVI #1,DELTA_E	
0xA2	906	GongTest	CMPI #0,DELTA_E	
0xA3	DA6		BRC Ton	
0xA4	905		CMPI #0,DELTA_D	Nichts gewonnen?
0xA5	EA8		BRZ GewinnAnzeige	Dann nur anzeigen
0xA6	70F	Ton	SUBI #0,rF	Sound erzeugen (spez. OpCode)
0xA7	000		MOV r0,r0	Sound stoppen (spez. OpCode)
0xA8	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA9	F3C		DISP #3,GELD_PF_EINER	
0xAA	BF1		CALL Warteschleife	setzt auch R0 auf 0 wg. EXRA
0xAB	F02		DISOUT	
0xAC	F0F		EXRA	
0xAD	BEA		CALL DeltaAdd	
0xAE	BFC		CALL LöscheDelta	Geldgewinne löschen
0xAF	90A		CMPI #0,STATUS	Laufen keine Sonderspiele?
0xB0	EC9		BRZ SoGewinn	dann EFE oder FFF testen

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xB1	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xB2	1F5		MOVI #F,DELTA_D	vorher -1 Sonderspiel (akt. Spiel)
0xB3	BE8		CALL DeltaSub	
0xB4	BFC		CALL LöscheDelta	
0xB5	7F2		SUBI #F,MI	Tableau testen: König in der Mitte?
0xB6	DC8		BRC EndeTableau	kein König in der Mitte
0xB7	90E	Tableau	CMPI #0,S_ZEHNER	Pos > 16?
0xB8	DC8		BRC EndeTableau	dann keine Tableaugewinne
0xB9	99D		CMPI #9,S_EINER	Pos > 10?
0xBA	DC2		BRC Pos16	dann Pos16 und Pos15 prüfen
0xBB	EC0		BRZ Gewinn10S	Pos = 10? dann 10 Sonderspiele
0xBC	125		MOVI #2,DELTA_D	Pos < 10? dann mind. 2 Sonderspiele
0xBD	90D		CMPI #0,S_EINER	Pos = 1?
0xBE	DC8		BRC EndeTableau	
0xBF	585	Gewinn20S	ADDI #8,DELTA_D	20 Sonderspiele geben:
				2 Sonderspiele aufstocken auf 10
				und zusätzlich...
0xC0	5A5	Gewinn10S	ADDI #A,DELTA_D	10 Sonderspiele geben
0xC1	FB6		ADC DELTA_E	
0xC2	9ED	Pos16	CMPI #E,S_EINER	Pos > 15?
0xC3	DC6		BRC Gewinn50S	Dann 50 Sonderspiele
0xC4	EC7		BRZ Gewinn2S	Pos = 15? dann 2 Sonderspiele
0xC5	CC8		GOTO EndeTableau	
0xC6	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele (Hex 32)
0xC7	125	Gewinn2S	MOVI #2,DELTA_D	
0xC8	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xC9	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >15 Sonderspiele gewonnen?
0xCA	DD5		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xCB	9A5		CMPI #A,DELTA_D	Genau 10 gewonnen?
0xCC	ED5		BRZ SoAnzeige	dann anzeigen
0xCD	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xCE	DD3		BRC ZehnS	
0xCF	ED1		BRZ FünfS	
0xD0	CD5		GOTO SoAnzeige	
0xD1	155	FünfS	MOVI #5,DELTA_D	5 Sonderspiele geben
0xD2	CD4		GOTO SoGewonnen	
0xD3	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben
0xD4	12A	SoGewonnen	MOVI #2,STATUS	Statusregister auf "Sonderspiele" setzen
0xD5	90A	SoAnzeige	CMPI #0,STATUS	Keine Sonderspiele?
0xD6	EE5		BRZ Aufräumen	dann keine Anzeige und aufräumen
0xD7	F0F		EXRA	Sonderspiele-Gewinn anzeigen
0xD8	F2D		DISP #2,S_EINER	
0xD9	BF1		CALL Warteschleife	
0xDA	F0F		EXRA	
0xDB	F02		DISOUT	
0xDC	F0E		EXRM	Sonderspiele-Register einblenden

